

I'm Leaving on an *Overbooked* Jet Plane

Jen Bergman, Doug Hogan, Martin Schettler; Juniata College Mathematics Department

Contents

Introduction	1
Variables.....	2
Equation of Model.....	2
Assumptions.....	3
Code for the max_profit Procedure.....	4
Initial Plots	5
Initial State Example.....	6
Effects of Changes in Parameters on the Model	8
Effects of the September 11th Tragedy on the Model	9
Conclusion.....	11
Appendix A -1 -- The Initial Model (Maple Worksheets).....	12
Appendix A -2 -- Effects of Changes on the Model (Maple Worksheets).....	15
Appendix B - Memo to Airline CEO	25
Works Cited.....	26

Introduction

Because not all passengers fly on the flight that they initially schedule, airlines overbook their flights in order to maximize their profit. This practice forces airlines to balance the cost of overbooking (when more passengers than expected arrive) with the increased revenue of receiving payment from passengers who do not end up taking their flights. Profits are calculated by subtracting the costs to the airline from the total revenue generated by ticket sales.

The costs can be broken down into two cases. When there are enough seats on the plane for all of the customers who try to board, the costs are all grouped into a variable we shall call airline overhead costs. If more passengers than expected arrive, the cost increases to account for compensating “bumped” customers.

Our model considers the profits of airlines in one week. We analyze profit in terms of two variables: the fraction by which the airlines will overbook flights and the fraction of customers who actually board the plane. We create a model based on the situation before the September 11, 2001 attacks, then we consider the effects of these attacks on the airline industry. In all cases, we are looking for the fraction by which airlines should overbook flights in order to maximize their profits.

We will do the bulk of our analysis numerically using the Waterloo Maple 6 software package. We define an equation for profit that represents the model, and we use Maple to calculate how each of our variables affects the profit. In addition, Maple becomes helpful

in creating graphs that show what happens when the variables change. We use a numerical procedure to look at all plausible values of each of the variables when calculating the optimal overbooking fraction. This procedure also plots various values of the overbooking fraction against the average profits made.

Variables

In order to analyze profit in relation to overbooking, we will create a model to be used as a general case. This model expresses profit per week, P , as a function of the predicted fraction of seats to book based on the expected fraction of ticket-holders who actually enplane, q , and the fraction of ticket-holders who actually enplane, b . This means that the predicted number of ticket-holders that will enplane is $1/q$ and that the fraction overbooked is $(q-1)$. S is the number of seats on a plane, which means that the number of tickets sold for each flight is qS . Other variables used are the number of flights each week, n , the average ticket price, t , and the compensation given to each passenger who is unable to board their flight due to overbooking and too many passengers, C . A final variable considered is airline overhead costs, A . This variable includes pilot, captain, and flight attendant salaries, supplies for flights, airplane fees, security, etc.

Equation of Model

Combining all of these variables, we are able to obtain a model that represents profit in relation to these variables and overbooking. In our initial model we consider S , n , t , C , and A to be constants. Following this initial model and using our calculating program, we consider the effects of changes to and variations of these constant values.

To derive our model, we must first consider the idea that $profit = revenue - cost$.

Revenue is defined as the number of flights in a week multiplied by the number of seats on a plane, the fraction of seats booked, and the cost of each ticket. Therefore,

$$revenue = nSqt.$$

Cost is defined as the sum of the overhead and compensation fees. The overhead fees are A . $1/q$ represents the predicted fraction of tickets bought that will be used. b represents the fraction of ticket-holders who use their tickets, therefore, $b - (1/q)$ is the number of ticket-holders who receive compensation. If $b = 1/q$, then the plane is filled and no compensation is given to ticket-holders. With $b > 1/q$ more ticket-holders try to enplane than there are seats, resulting in a compensation cost being spent. If $b < 1/q$ then no compensation cost is given and the profit is simply the revenue with the overhead costs subtracted. This is also true for $b = 1/q$. If $b > 1/q$ then the compensation cost is $(b-1/q)$ or the fraction of ticket-holders who receive compensation multiplied by the number of tickets sold (qS), the number of flights in a week, and the actual compensation fee for each ticket-holder.

Given these variables and ideas, the model we derive is:

$$P(q,b) = \begin{cases} nSqt - \left[A + ncSq \left(b - \frac{1}{q} \right) \right], & b \geq \frac{1}{q} \\ nSqt - A, & b < \frac{1}{q} \end{cases}$$

This model calculates P , profit, as a function of q and b , which are respectively the fraction of seats booked and the fraction of ticket-holders who enplane. The other variables are all set as constants and various cases are considered.

Assumptions

In order to make this project manageable, we need to simplify the situation and set multiple assumptions. In considering changes that could be made to the model to make it more effective, these assumptions should be accounted for in the model. By adding and changing variables, it would be possible to explain these ideas and accommodate for them within the model, making it more realistic. Some of our assumptions involve set values for our constants. These values mirror those of actual airplanes and airlines.

We assume that at least Sq tickets will be sold. This means that the number of tickets sold is equal to the fraction we predicted times the number of seats on a plane.

q is greater than or equal to one. This leads to the idea that the number of tickets sold is equal to or greater than the number of seats on a plane.

No refunds are offered. For our simplified model, those who do not enplane do not receive refunds for the tickets that they did not use.

10-15 percent of passengers who buy tickets do not enplane. This makes our range of values of b defined from 0.85 to 0.90 (Khan).

We are considering all planes to be 747s. This makes our S , or number of seats, 416 for each plane ("Boeing 747 Technical Specifications").

There are 10500 flights per week (United Airlines).

The compensation fee is a minimum of \$400 (Khan). Our model uses a value of \$500 in order to consider the situation in a more extreme case.

Code for the max_profit Procedure

This Maple procedure is used to find the maximum profit and the value of q , the fraction of seats to sell tickets, for which profit is maximized:

```
max_profit:=proc(min_q1,max_q1,increment,show_pts)

# pre: max_q1 > min_q1 >= 1 && increment > 0 &&
#       show_pts == 0 or 1
# post: calculate the average profit for each value of q (fraction by which
#       the plane is booked) for each value of b (fraction of ticketholders who
#       board) and compute the q for which the average profit is maximized;
#       report this q and its corresponding profit;
#       show_pts == 1 --> display the average profit for each q;
#       a global variable ptsplot has been defined and the "display(ptsplot);"
#       command can follow this function to display a graph of the q vs.
#       the average profit for q

global ptsplot;           # plot of q vs. average profit
local min_b1, max_b1,    # extremes for fraction of ticketholders who board
      max_index,        # value of q for which profit is maximized
      array_size,      # how many data points to store
      amonkey,         # array to store all data points (q, avg profit)
      i,               # loop control variable/index to the array
      q1, b1,         # LCVs for varying q and b
      sum1, num,      # accumulators for finding average profits
      average;        # average profit for each value of q

min_b1 := infinity;      # set up to find extremes
max_b1 := 0:

max_index := 0:

array_size := floor((max_q1 - min_q1)/increment):  # calculate how many points

amonkey := array(0..array_size):                 # create storage for data pts
i := 0:                                           # set up LCV

for q1 from min_q1 to max_q1 by increment        # vary fraction of seats booked
do
  sum1 := 0:                                     # set up to calculate averages
  num := 0:

  for b1 from 0.85 to 0.9 by 0.005              # vary % of boarding passengers
  do                                             # based on data from (Khan)
    sum1 := sum1 + P(q1,b1):                   # tally values
    num := num + 1:

  od:
  average := sum1/num;                          # calculate average

  amonkey[i] := [q1,average];                  # store average
  i := i + 1;                                  # increment LCVs for next data pt

  if(average < min_b1) then                    # track minimum
    min_b1 := average;
  fi;

  if(average > max_b1) then                    # track maximum
    max_b1 := average;
    max_index := q1;
  fi;

  if(show_pts=1) then                          # check whether or not to display
    lprint(q1, average);                      # table of all average profits
  fi;


od:

ptsplot := plot(amonkey,style=point,
```


```
view=[min_q1..max_q1,(min_b1-1000)..(max_b1+1000)],  
symbol=circle, color = blue); # set up a plot of the graph  
# that can be displayed  
lprint("max profit at q= ", max_index); # report extrema  
lprint("max profit: ", max_b1);  
  
end;
```

Initial Plots of the Model


To get an idea of where our values of q will be that will maximize profit, we begin by fixing b at three values and plotting the two dimensional plot of $P(q,b)$ with b fixed. These graphs follow and show that our model follows common sense in that the more people that actually arrive to board the flight, the fewer number of tickets we should overbook. Note that the commands to produce these and all graphs appearing in this paper are included in Appendix A.




This graph has b set at 0.99. In other words, this is a graph of P as a function of q assuming that 99% of passengers with tickets will take the flight.



This graph has $b = 0.9$, and shows that with this lower value of b allows us to overbook at a greater rate. The following graph shows this in the more extreme case where only 80% of passengers with tickets take the flight.




To get a better idea of the behavior of our model in three-space, we show two three-dimensional plots of $P(q,b)$.




Initial State Example

In order to use our model, we assign it some initial parameters. For our initial state, we set $S = 416$, $t = 271.28$, $A = 250,000,000$, $C = 500$, and $n = 10500$. With these constants, we run our `max_profit` procedure (with parameters 1.0,1.30,0.01,0) to get our maximum profit \$1,088,024,424, at $q = 1.15$. The procedure also gives us the following plot, which provides graphical confirmation of these values.



Using this initial run of the procedure as a guide, we narrow our domain for the next call to the region from $1.14 < q < 1.16$, and also use the smaller step-size of 0.001 for the calculations. Calling the procedure again with these parameters gives us a maximum

profit of \$1,238,024,425 at $q = 1.150$. Again, this can be verified from the following graph.



Effects of Changes in Parameters on the Model

Here we will investigate the effects of changing each of the parameters independently of the others. We report the resulting maximum profits here and their corresponding values of q .

For details on the commands used to calculate these values and the exact changes in the parameters, see Appendix A-2.

Parameter Change		New Maximum Profit	Corresponding q
Initial State		\$1,238,024,425	1.150
Number of seats			
	Increased	\$1,679,842,918.00	1.150
	Decreased	\$554,101,215.90	1.150
Ticket prices			
	Increased	\$1,486,109,251.00	1.163
	Decreased	\$483,605,600.00	1.130
Number of flights			
	Increased	\$1,470,317,115.00	1.150
	Decreased	705,731,730.90	1.150
Airline overhead costs			
	Increased	\$838,024,423.30	1.150
	Decreased	\$1,213,024,424.00	1.150
Compensation for bumped customers			
	Increased	\$1,083,795,976.00	1.143
	Decreased	\$1,094,589,940.00	1.163

Note that all of these changes in relation to the initial state example are reasonable. For example, considering our model, no change to q occurs when airline overhead costs, A , are changed. This change does not affect the overbooking, only the amount of profit made.

Effects of September 11 Tragedy on the Overbooking Model

We make the following assumptions on the effects of the September 11 tragedy on the variables of our model. Given these assumptions, we run our procedure to observe their effects on the maximum profit, and find the optimal overbooking strategies in light of the recent tragedy.

Fewer Flights: Because there are fewer flights, the number of flights, n , decreases. Because there are fewer flights, there will be fewer pilots who need to be paid, less food that will need to be served, and less fuel that will need to be purchased, meaning that less money must be paid in airport fees. Thus, lowering the number of flights also brings about a decrease in the airline overhead cost, A .

Heightened Security: The September 11 tragedy has resulted in heightened security at airports and on planes themselves. In our model, security falls in the category of airline overhead costs, so A will increase. In an ABC News article, it is reported that increased security will result in increased ticket prices for the customers. This increase includes a surcharge of about \$2.50 per ticket, as proposed by the Senate (Dizikes). Thus, t will increase as a result of this factor.

Increased Passenger Fear: Because people are scared to fly as a result of the September 11 events, the airline industry is spending more money on public relations campaigns to remind consumers that flying is safe. This advertising results in an increase in the airline overhead cost A . In addition, since fewer people feel safe flying, the airline industry does not offer as many flights or cannot afford to do so without losing money. Thus, the number of flights, n , decreases as a result of this factor. Smallen reported there were 18 percent fewer flights in October 2001 than there were in October 2000, so we will use this statistic in our calculations.

Loss of Revenue by Airline Industry: Because the airline industry has already lost so much money as a result of September 11, it will try to compensate for its lost revenue. This can be done by increasing ticket prices, t . Also, airlines will be less willing to compensate customers who have been “bumped” from their seats, so C will decrease as a result of the airline industry losing revenue and trying to save on costs.


Summarizing the results of these four factors, we will change the parameters to our model as follows:

- n will **decrease**.
 - We will reduce it by 18 percent, so it will change from 10,500 to 8600.
- A will **increase**.
 - However, the change in A is arbitrary. It is simply a vertical shift in the profit curve. Since we are not concerned with the actual maximum profit as much as what value of q gives us the maximum profit, it does not matter how much A changes.
 - We will leave A unchanged since its value does not affect our results.
- T will **increase**.


- We've chosen to increase it from \$271.28 to \$314.15.
- **C will decrease.**
 - We will reduce the compensation for a customer who is bumped from \$500 to \$400.

When we apply these changes to our model, we do not expect a change to the initial model results as a result of the number of flights decreasing. The overbooking fraction will not be affected. Considering what we have seen by changing the parameters before, the increase in the price of tickets should bring about an increase in q . In addition, decreasing the compensation rate, C , should also increase q . Thus, our overall prediction is that the September 11 tragedy will cause an increase in the value of q .

When we run our procedure with these new parameters, we get the following plot of the average value of q vs. the profit:



When we zoom in on the maximum, we see the following plot:



Our procedure and these graphs tell us a maximum profit of \$1,030,263,837 is obtained when q is 1.170. Thus, in light of the September 11 events, our model suggests that airlines should overbook flights by **17 percent** in order to maximize their profits.

As predicted, we also observe here that profits have gone down as a result of September 11. Our model shows a decrease in profits from \$1,238,024,425 before the tragedy to \$1,030,263,837 afterward. However, these profit values are not terribly meaningful since our data are not exact. Nevertheless, we saw a decrease in profits, which was expected.

Conclusion

By applying our numerical procedure to the profit model we generated, we were able to compute the optimal fraction by which an airline should overbook its flights. In our initial model, where we analyzed the state of the airline industry before September 11, we found that overbooking by 15 percent gave the greatest profits. For a Boeing 747 jet, this fraction means that 62 more tickets than seats should be sold. Obviously, our model makes several assumptions, whereby the results of our analysis are slightly weakened. Additionally, stronger empirical data would increase the reliability of our scheme.

We also investigated the effects of the September 11 tragedy on the overbooking strategy airline industry. As a result of fewer flights being offered, passengers being afraid to fly, heightened security, and loss of revenue, our model suggests that increasing the overbooking factor would be in an airline's best interests. The increase we calculated was approximately two percent.

Appendix A -1 -- The Initial Model

The following are all the commands used for the Initial State Example section.

Variables:

P = profit per week

t = average ticket price

A = airline overhead cost (pilot salaries, flight attendant salaries, supplies for flights, security, etc.)

b = fraction of ticketholders who actually enplane

q = predicted fraction of seats to overbook based on expected **b**

(1/q) = predicted number of ticketholders that will enplane

S = number of seats per plane

(S*q) = number of tickets sold per plane

C = compensation cost for a passenger who gets bumped

n = number of flights per week

Profit = Revenue - Cost

= (ticket cost * number of tickets * number of flights) - (airline overhead + compensation for overbooking)

Two cases:

1. More people than there are seats will enplane. The airline will have to compensate ticketholders who are bumped, so there will be additional costs beyond flying the plane.

2. There will be enough seats for the people who enplane (or there will be extra seats). In this case, the airline will not have any costs beyond flying the plane.

```
> restart; with(plots):
```

```
Warning, the name changecoords has been redefined
```


```
> P := (q,b) -> piecewise(b>1/q, t*S*q*n - (A + n*C*(b-1/q)*S*q),
                          b<=1/q, t*S*q*n - A);
```

$$P := (q, b) \rightarrow \text{piecewise}\left(\frac{1}{q} < b, t S q n - A - n C \left(b - \frac{1}{q}\right) S q, b \leq \frac{1}{q}, t S q n - A\right)$$


Looking at an example....

```
> S := 416;           # seats on the plane
> t := 271.28;       # an average ticket cost we're picking
> A := 250000000;    # airline overhead costs
> C := 500;          # compensate bumped customer with $500 (assuming
the largest value)
n := 10500;          # number of flights per week
                      S := 416
                      t := 271.28
                      A := 250000000
                      C := 500
                      n := 10500
```


```
> plot(P(q,.9), q=1.00..1.50, title="P(q,.9), q=1.00..1.50");
```




```
> plot(P(q,.99), q=1.00..1.50, title="P(q,.99), q=1.00..1.50");  
P(q,.99), q=1.00..1.50
```




```
> plot(P(q,.8), q=1.00..1.50, title="P(q,.8), q=1.00..1.50");  
P(q,.8), q=1.00..1.50
```




```
> plot3d(P(q,b), q=1.00..2.00, b=0..1);
```




```
> plot3d(P(q,b), q=1..2, b=0.5..1);
```




```
> plot3d(P(q,b), q=1..1.2, b=0.8..1, title="P(q,b)");  
P(q,b)
```



```
> read(`h:/max_profit.txt`):  
> max_profit(1.0,1.30,0.01,0);  
display(ptsplot);  
"max profit at q= ", 1.15  
"max profit: ", 1088024424.
```



```
> max_profit(1.14,1.16,0.001,0);  
display(ptsplot);  
"max profit at q= ", 1.150  
"max profit: ", 1238024425.
```



Appendix A -2 -- Effects of Changes on the Model

The following are all the commands used to compile the table in the **Effects of Changes in Parameters of the Model** section. The process is essentially the same for each change; we either increase or decrease one of the parameters, and run the max_profit procedure to obtain the graph of data points and the maximum profit. Below is also how the **Effects of the September 11th Tragedy on the Model** section was created and analyzed.


Increased number of seats

```
> restart; with(plots):
Warning, the name changecoords has been redefined


> P := (q,b) -> piecewise(b>1/q, t*S*q*n - (A + n*C*(b-1/q)*S*q),
                        b<=1/q, t*S*q*n - A);
      P := (q, b) -> piecewise( $\frac{1}{q} < b, t S q n - A - n C \left( b - \frac{1}{q} \right) S q, b \leq \frac{1}{q}, t S q n - A$ )

> S := 600;           # seats on the plane
> t := 271.28;       # an average ticket cost we're picking
> A := 250000000;    # airline overhead costs
> C := 500;         # compensate bumped customer with $500 (assuming
                    # the largest value)
n := 10500;         # number of flights per week
                    S := 600
                    t := 271.28
                    A := 250000000
                    C := 500
                    n := 10500

> read(`h:/max_profit.txt`):
> max_profit(1.0,1.30,0.01,0);
display(ptsplot);
"max profit at q= ", 1.15
"max profit: ", 1679842918.
```



```
> max_profit(1.14,1.16,0.001,0);
display(ptsplot);
"max profit at q= ", 1.150
"max profit: ", 1679842918.
```



Decreased number of seats


```
> restart; with(plots):
Warning, the name changecoords has been redefined


> P := (q,b) -> piecewise(b>1/q, t*S*q*n - (A + n*C*(b-1/q)*S*q),
                          b<=1/q, t*S*q*n - A);
      P := (q, b) -> piecewise( $\frac{1}{q} < b, t S q n - A - n C \left( b - \frac{1}{q} \right) S q, b \leq \frac{1}{q}, t S q n - A$ )

> S := 250;           # seats on the plane
> t := 271.28;       # an average ticket cost we're picking
> A := 250000000;    # airline overhead costs
> C := 500;         # compensate bumped customer with $500 (assuming
                    # the largest value)
n := 10500;         # number of flights per week
                    S := 250
                    t := 271.28
                    A := 250000000
                    C := 500
                    n := 10500

> read(`h:/max_profit.txt`):
> max_profit(1.0,1.30,0.01,0);
display(ptsplot);
"max profit at q= ", 1.15
"max profit: ", 554101215.9

> max_profit(1.14,1.16,0.001,0);
display(ptsplot);
"max profit at q= ", 1.150
"max profit: ", 554101215.9
```






Increased ticket prices

```
> restart; with(plots):
Warning, the name changecoords has been redefined


> P := (q,b) -> piecewise(b>1/q, t*S*q*n - (A + n*C*(b-1/q)*S*q),
                          b<=1/q, t*S*q*n - A);
      P := (q, b) -> piecewise(1/q < b, t S q n - A - n C (b - 1/q) S q, b <= 1/q, t S q n - A)

> S := 416;           # seats on the plane
> t := 350;           # an average ticket cost we're picking
> A := 250000000;     # airline overhead costs
> C := 500;           # compensate bumped customer with $500 (assuming
the largest value)
n := 10500;          # number of flights per week
                    S := 416
                    t := 350
                    A := 250000000
                    C := 500
                    n := 10500

> read(`h:/max_profit.txt`):
> max_profit(1.0,1.30,0.01,0);
display(ptsplot);
"max profit at q= ", 1.16
"max profit: ", 1485763782.
```



```
> max_profit(1.15,1.17,0.001,0);
display(ptsplot);
"max profit at q= ", 1.163
"max profit: ", 1486109251.
```




Decreased ticket prices

```
> restart; with(plots):
Warning, the name changecoords has been redefined


> P := (q,b) -> piecewise(b>1/q, t*S*q*n - (A + n*C*(b-1/q)*S*q),
                        b<=1/q, t*S*q*n - A);
P := (q, b) -> piecewise( $\frac{1}{q} < b, t S q n - A - n C \left( b - \frac{1}{q} \right) S q, b \leq \frac{1}{q}, t S q n - A$ )

> S := 416;           # seats on the plane
> t := 150;          # an average ticket cost we're picking
> A := 250000000;    # airline overhead costs
> C := 500;          # compensate bumped customer with $500 (assuming
the largest value)
n := 10500;         # number of flights per week
                    S := 416
                    t := 150
                    A := 250000000
                    C := 500
                    n := 10500
```

```
> read(`h:/max_profit.txt`):
> max_profit(1.0,1.30,0.01,0);
display(ptsplot);
"max profit at q= ", 1.13
"max profit: ", 483605600.0
```



```
> max_profit(1.12,1.14,0.001,0);
display(ptsplot);
"max profit at q= ", 1.130
"max profit: ", 483605600.0
```



Increased number of flights

```
> S := 416;           # seats on the plane
```

```
> t := 271.28;      # an average ticket cost we're picking
> A := 250000000;  # airline overhead costs
> C := 500;        # compensate bumped customer with $500 (assuming
the largest value)
n := 13500;        # number of flights per week
                    S := 416
```


t := 271.28

A := 250000000


C := 500

n := 13500

```
> read(`h:/max_profit.txt`):
> max_profit(1.0,1.30,0.01,0);
display(ptsplot);
"max profit at q= ", 1.15
"max profit: ", 1470317115.
```



```
> max_profit(1.14,1.16,0.001,0);
display(ptsplot);
"max profit at q= ", 1.150
"max profit: ", 1470317115.
```



Decreased number of flights

```
> S := 416;        # seats on the plane
> t := 271.28;     # an average ticket cost we're picking
> A := 250000000; # airline overhead costs
> C := 500;        # compensate bumped customer with $500 (assuming
the largest value)
n := 7500;         # number of flights per week
                    S := 416
```


t := 271.28

A := 250000000


C := 500

n := 7500

```
> read(`h:/max_profit.txt`):  
> max_profit(1.0,1.30,0.01,0);  
display(ptsplot);  
"max profit at q= ", 1.15  
"max profit: ", 705731730.9
```




```
> max_profit(1.14,1.16,0.001,0);  
display(ptsplot);  
"max profit at q= ", 1.150  
"max profit: ", 705731730.9
```




Increased airline overhead costs

```
> S := 416;           # seats on the plane  
> t := 271.28;       # an average ticket cost we're picking  
> A := 500000000;   # airline overhead costs  
> C := 500;         # compensate bumped customer with $500 (assuming  
the largest value)  
n := 10500;         # number of flights per week  
                    S := 416  
                    t := 271.28  
                    A := 500000000  
                    C := 500  
                    n := 10500  
  
> read(`h:/max_profit.txt`):  
> max_profit(1.0,1.30,0.01,0);  
display(ptsplot);  
"max profit at q= ", 1.15  
"max profit: ", 838024423.3
```



```
> max_profit(1.14,1.16,0.001,0);  
display(ptsplot);  
"max profit at q= ", 1.150  
"max profit: ", 838024423.3
```



Decreased airline overhead costs

```
> S := 416;           # seats on the plane  
> t := 271.28;       # an average ticket cost we're picking  
> A := 125000000;   # airline overhead costs  
> C := 500;         # compensate bumped customer with $500 (assuming  
the largest value)  
n := 10500;         # number of flights per week
```

S := 416


t := 271.28

A := 125000000


C := 500

n := 10500

```
> read(`h:/max_profit.txt`):  
> max_profit(1.0,1.30,0.01,0);  
display(ptsplot);  
"max profit at q= ", 1.15  
"max profit: ", 1213024424.
```



```
> max_profit(1.14,1.16,0.001,0);  
display(ptsplot);  
"max profit at q= ", 1.150  
"max profit: ", 1213024424.
```



Increased compensation

```
> S := 416;           # seats on the plane
> t := 271.28;        # an average ticket cost we're picking
> A := 250000000;    # airline overhead costs
> C := 600;           # compensate bumped customer with $500 (assuming
the largest value)
n := 10500;          # number of flights per week
```

```
S := 416
```


```
t := 271.28
```

```
A := 250000000
```


```
C := 600
```

```
n := 10500
```

```
> read(`h:/max_profit.txt`):
> max_profit(1.0,1.3,0.01,0);
display(ptsplot);
"max profit at q= ", 1.14
"max profit: ", 1083451605.
```



```
> max_profit(1.13,1.15,0.001,0);
display(ptsplot);
"max profit at q= ", 1.143
"max profit: ", 1083795976.
```




Decreased compensation


```
> S := 416;           # seats on the plane
> t := 271.28;        # an average ticket cost we're picking
```

```
> A := 250000000; # airline overhead costs
> C := 400; # compensate bumped customer with $500 (assuming
the largest value)
n := 10500; # number of flights per week
S := 416
t := 271.28
A := 250000000
C := 400
n := 10500

> read(`h:/max_profit.txt`):
> max_profit(1.0,1.3,0.01,0);
display(ptsplot);
"max profit at q= ", 1.16
"max profit: ", 1094427832.
```



```
> max_profit(1.15,1.17,0.001,0);
display(ptsplot);
"max profit at q= ", 1.163
"max profit: ", 1094589940.
```




Effects of September 11


```
> S := 416; # seats on the plane
> t := 314.15; # an average ticket cost we're picking
> A := 250000000; # airline overhead costs
> C := 400; # compensate bumped customer with $500 (assuming
the largest value)
n := 8600; # number of flights per week (K)
S := 416
t := 314.15
A := 250000000
C := 400
n := 8600

> read(`h:/max_profit.txt`):
```

```
> max_profit(1.0,1.3,0.01,0);  
display(ptsplot);  
"max profit at q= ", 1.17  
"max profit: ", 1030263837.
```



```
> max_profit(1.16,1.18,0.001,0);  
display(ptsplot);  
"max profit at q= ", 1.170  
"max profit: ", 1030263837.
```



Appendix B – Memo to CEO

MEMORANDUM

TO: LARRY AIRLINE, CEO
FROM: MCM
SUBJECT: OPTIMAL OVERBOOKING STRATEGY
DATE: 5/3/02

In light of the current condition of the airline industry, we have found it in our best interest to re-examine our overbooking strategy. We have developed a model of our overbooking strategy based on costs, ticket prices, passenger tendencies, and the number of flights we fly per week. Through rigorous analysis of our situation we have concluded that the optimal overbooking strategy in order to maximize profit is to increase overbooking to approximately 17% per plane. While this alone is unlikely to solve our current difficulties, it is a step in the right direction.

Sincerely,
MCM

Works Cited

- “2000 Pilot Salary Survey – Summary Results.” *AVCrew.com*. 2000. 8 Feb. 2002
<<http://www.avcrew.com/salary2000.htm>>.
- “Boeing 747 Technical Specifications.” *Boeing.com*. 22 Jan. 2002. 9 Feb. 2002
<<http://www.boeing.com/commercial/747family/technical.html>>.
- Dizikes, Peter. “Security Costs: Who Will Pay for Airline Safety?” *ABCNews.com*. 4 Oct.
2001. 9 Feb. 2002 <<http://abcnews.go.com>>.
- Jackson, Brian. “Airline ‘Bumping’ Up; Delta Leads List in Overbooking.” *Chicago Sun
Times* 8 July 1999: 52. General News. Lexis-Nexis Academic Universe. L.A. Beeghly
Library, Huntingdon, PA. 8 Feb. 2002 <<http://web.lexis-nexis.com/>>.
- Khan, Salina. “More Fliers Getting Left at Gate; Overbooking Raises Bumping of
Passengers.” *USA Today* 19 Apr. 1999: 3B. General News. Lexis-Nexis Academic
Universe. L.A. Beeghly Library, Huntingdon, PA. 8 Feb. 2002 <[http://web.lexis-
nexis.com/](http://web.lexis-
nexis.com/)>.
- Office of Aviation Enforcement and Proceedings. *Air Travel Consumer Report*. U.S.
Dept. of Trans. July 2001. 8 Feb. 2002
<<http://www.dot.gov/airconsumer/atcr01.htm#2001>>.
- Salant, Jonathan D. “Feds Look at Airline Overbooking.” *The Detroit News* 17 Apr.
2001. 7 Feb. 2002 <<http://detnews.com/2001/business/0104/17/c01-213233.htm>>.
- Smallen, David. “Impact of September 11 on Airlines.” Bureau of Transportation
Statistics. 9 Jan. 2002. 8 Feb. 2002
<<http://www.bts.gov/PressReleases/2002/bts0102.html>>.
- Taylor, Doug, ed. “Newsletter.” *JetCareers.com*. 28 Nov. 2000. 8 Feb. 2002
<<http://www.jetcareers.com/newsletter1128.htm>>.
- United Airlines. “United Airlines Issues Details of International Operational Plan;
Confirms Plans for Increased Domestic Flight Schedules for Next Week.” 14 Sept.
2001. 9 Feb. 2002 <http://www.ual.com/site/primaryPR/0,10026,1534_890,00.htm>.